

# **PPGCA - CAIA 10 - Engenharia de Ontologias e Aplicações Semânticas**

**Profa. Dra Rita Berardi e Prof. Dr. Cesar Tacla**

**Nicolas Abril**

## **Avaliação Final**

**Domínio escolhido:** Atividades dos grupos PET.

Atualmente as informações sobre que atividades cada grupo PET fez são recolhidas separadamente por cada grupo, na forma que acham melhor ou mais conveniente. Uma vez por ano é enviado um relatório das atividades planejadas e executadas para serem avaliadas pela entidade competente na forma de um documento PDF com formato mais ou menos estruturado.

Essa falta de conhecimento sobre as atividades dos outros grupos dificulta colaborações, faz com que boas ideias e projetos bem sucedidos fiquem perdidos, desconhecidos para outros grupos PET e favorece a repetição desnecessária de pesquisas e trabalhos.

Um bom modelo das atividades utilizado por todos os grupos e acessível de forma aberta poderia dar mais visibilidade ao PET, ajudar na sua administração e aumentar o impacto das atividades realizadas.

A modelagem foi feita baseada na minha experiência de participação no grupo PET Engenharia de Computação da UTFPR (PETECO), com o auxílio do prof. Tacla que é o professor tutor do PETECO. Além disso, o modelo busca acomodar as informações contidas nos relatórios de atividade do grupo. Para a realização das buscas, eu alimentei o modelo com o relatório de atividades de 2019, primeiramente convertido de PDF para formato tabular e em seguida para o modelo usando OntoRefine.

Como os dados foram convertidos de forma mais ou menos direta, boa parte da ontologia não está sendo usada, seria necessário ler os textos do relatório um por um para preencher corretamente. Portanto, algumas perguntas não vão conseguir ser respondidas devido à falta de dados. Como as informações utilizadas no trabalho são referentes apenas ao PETECO, as perguntas serão formuladas de acordo, mais voltadas para a gestão do grupo, mas a ontologia possibilita análises maiores.

## Perguntas:

### 1. Quantas atividades o PETECO planejou para 2019 e quantas completou?

Planejadas:

```
select (count(?atividade_planejada) as ?count_planejada)
where {
    ?atividade_planejada rdf:type atipet:Atividade .
}
```

Concluídas:

```
select (count(?atividade_completa) as ?count_completa)
where {
    ?atividade_completa rdf:type atipet:Atividade ;
        atipet:temEstadoAtividade atipet:Concluída.
}
```

### 2. Quantas horas foram usadas planejando atividades que não foram executadas em 2019?

Query:

```
select (sum(?horas) as ?soma_horas)
where {
    ?atividade rdf:type atipet:Atividade ;
        atipet:temEstadoAtividade atipet:NãoIniciada .
    ?atividade atipet:horasTrabalhoAtividade ?horas .
}
```

### 3. Quais foram as atividades que estavam sendo realizadas no mês de setembro?

Query:

```
select ?atividade ?nome
where {
    ?atividade rdf:type atipet:Atividade .
    ?atividade atipet:títuloAtividade ?nome.
    ?atividade atipet:temPeríodoDeExecução ?periodo.
    ?periodo time:hasBeginning ?start;
        time:hasEnd ?end.
    ?start time:inDateTime ?start_desc.
    ?start_desc time:year ?year_start;
        time:month ?month_start.
    ?end time:inDateTime ?end_desc.
    ?end_desc time:year ?year_end;
        time:month ?month_end.
    bind(xsd:integer(xsd:string(?year_start)) as ?ys)
    bind(xsd:integer(xsd:string(?month_start)) as ?ms)
    bind(xsd:integer(xsd:string(?year_end)) as ?ye)
    bind(xsd:integer(xsd:string(?month_end)) as ?me)
    filter( (?ys < 2019) || ((?ys = 2019) && (?ms <= 09)) ).
    filter( (?ye > 2019) || ((?ye = 2019) && (?me >= 09)) ) .
}
```

#### 4. Qual atividade teve maior público participante?

Essa pergunta não consegue ser respondida com os dados, já que eles não têm informação de participantes. Ela está presente aqui mais como exemplo de coisa que poderia ser descoberta com a ontologia.

Query:

```
select ?ati (count(?participacao) as ?num_pub)
where {
    ?participacao rdf:type atipet:RelacionadorParticipanteAtividade ;
                  atipet:referenteAAtividade ?ati ;
                  atipet:participaçãoComPapel ?papel .
    ?papel rdf:type atipet:PapelPúblico.
    ?ati rdf:type atipet:Atividade ;
        atipet:títuloAtividade ?nome .
}
group by ?ati
order by desc(count(?participacao))
```

#### 5. Quantas atividades cada petiano ajudou a realizar?

Assim como na pergunta anterior, os dados usados não têm informação dos petianos.

Query:

```
select ?petiano (count(?participacao) as ?num_part)
where {
    ?participacao rdf:type atipet:RelacionadorRealizadorAtividade ;
                  atipet:referenteAoParticipante ?petiano.
}
group by ?petiano
order by asc(count(?participacao))
```